

BUGS

## Pumps: conjugate gamma-Poisson hierarchical model

George *et al* (1993) discuss Bayesian analysis of hierarchical models where the conjugate prior is adopted at the first level, but for any given prior distribution of the hyperparameters, the joint posterior is not of closed form. The example they consider relates to 10 power plant pumps. The number of failures  $x_i$  is assumed to follow a Poisson distribution

$$x_i \sim \text{Poisson}(\theta_i t_i) \quad i = 1, \dots, 10$$

where  $\theta_i$  is the failure rate for pump  $i$  and  $t_i$  is the length of operation time of the pump (in 1000s of hours). The data are shown below.

Pump	$t_i$	$x_i$
1	94.5	5
2	15.7	1
3	62.9	5
4	126	14
5	5.24	3
6	31.4	19
7	1.05	1
8	1.05	1
9	2.1	4
10	10.5	22

A conjugate gamma prior distribution is adopted for the failure rates:


$$\theta_i \sim \text{Gamma}(\alpha, \beta), \quad i = 1, \dots, 10$$

George *et al* (1993) assume the following prior specification for the hyperparameters  $\alpha$  and  $\beta$

$$\begin{aligned} \alpha &\sim \text{Exponential}(1.0) \\ \beta &\sim \text{Gamma}(0.1, 1.0) \end{aligned}$$

They show that this gives a posterior for  $\beta$  which is a gamma distribution, but leads to a non-standard posterior for  $\alpha$ . Consequently, they use the Gibbs sampler to simulate the required posterior densities.

*Graphical model for pump example:*



BUGS language for pump example:

```

model
{
  for (i in 1 : N) {
    theta[i] ~ dgamma(alpha, beta)
    lambda[i] <- theta[i] * t[i]
    x[i] ~ dpois(lambda[i])
  }
  alpha ~ dexp(1)
  beta ~ dgamma(0.1, 1.0)
}

```

**Data** ⇒ list(t = c(94.3, 15.7, 62.9, 126, 5.24, 31.4, 1.05, 1.05, 2.1, 10.5),  
 x = c( 5, 1, 5, 14, 3, 19, 1, 1, 4, 22), N = 10) ⇐

**Inits** ⇒ list(alpha = 1, beta = 1) ⇐

## Results

A burn in of 1000 updates followed by a further 10000 updates gave the parameter estimates:

<b>node</b>	<b>mean</b>	<b>sd</b>	<b>MC error</b>	<b>2.5%</b>	<b>median</b>	<b>97.5%</b>	<b>start</b>	<b>sample</b>
alpha	0.7001	0.2699	0.004706	0.2851	0.6634	1.338	1001	10000
beta	0.929	0.5325	0.00978	0.1938	0.8315	2.205	1001	10000
theta[1]	0.0598	0.02542	2.68E-4	0.02128	0.05627	0.1195	1001	10000
theta[2]	0.1008	0.07855	8.177E-4	0.00838	0.08181	0.3023	1001	10000
theta[3]	0.08927	0.03759	3.702E-4	0.0316	0.08469	0.1762	1001	10000
theta[4]	0.116	0.03048	3.17E-4	0.06363	0.1132	0.1825	1001	10000
theta[5]	0.6056	0.315	0.003087	0.1529	0.5529	1.359	1001	10000
theta[6]	0.6105	0.1393	0.0014	0.3668	0.5996	0.9096	1001	10000
theta[7]	0.9025	0.7252	0.007937	0.07559	0.7167	2.751	1001	10000
theta[8]	0.8964	0.725	0.008262	0.07614	0.7098	2.785	1001	10000
theta[9]	1.59	0.7767	0.009004	0.4828	1.452	3.452	1001	10000
theta[10]	1.993	0.4251	0.004915	1.264	1.958	2.916	1001	10000